

# A Tutorial on Hidden Markov Models using Stan

*Luis Damiano (Universidad Nacional de Rosario), Brian Peterson (University of Washington), Michael Weylandt (Rice University)*

2017-12-15

## Contents

<b>1</b>	<b>The Hidden Markov Model</b>	<b>1</b>
1.1	Model specification	2
1.2	The generative model	3
1.3	Characteristics	4
1.4	Inference	5
1.5	Parameter estimation	10
1.6	A walk-through	11
<b>2</b>	<b>The Input-Output Hidden Markov Model</b>	<b>14</b>
2.1	Definitions	14
2.2	Inference	15
2.3	Parameter estimation	17
2.4	A walk-through the IOHMM	18
<b>3</b>	<b>Acknowledgements</b>	<b>24</b>
<b>4</b>	<b>Original Computing Environment</b>	<b>24</b>
<b>5</b>	<b>References</b>	<b>26</b>

This case study documents the implementation in [Stan](#) (Carpenter et al. 2016) of the Hidden Markov Model (HMM) for unsupervised learning (Baum and Petrie 1966; Baum and Eagon 1967; Baum and Sell 1968; Baum et al. 1970; Baum 1972). Additionally, we present the adaptations needed for the Input-Output Hidden Markov Model (IOHMM). IOHMM is an architecture proposed by Bengio and Frasconi (1995) to map input sequences, sometimes called the control signal, to output sequences. Compared to HMM, it aims at being especially effective at learning long term memory, that is when input-output sequences span long points. In all cases, we provide a fully Bayesian estimation of the model parameters and inference on hidden quantities, namely filtered state belief, smoothed state belief and jointly most probable state path.

*A Tutorial on Hidden Markov Models using Stan* is licensed under CC-BY 4.0. See the [LICENSE](#) file for details. All files are available in the [stancon18](#) GitHub repository.

---

## 1 The Hidden Markov Model

Real-world processes produce observable outputs characterized as signals. These can be discrete or continuous in nature, be pure or contaminated with noise, come from a stationary or non stationary source, among many other variations. These signals are modeled to allow for both theoretical descriptions and practical applications. The model itself can be deterministic or stochastic, in which case the signal is well characterized as a parametric random process whose parameters can be estimated in a well-defined manner.

Autocorrelation, a key feature in most signals, can be modeled in countless forms. While certainly pertinent to this purpose, high order Markov chains can prove inconvenient when the range of the correlation among

the observations is long. A more parsimonious approach assumes that the observed sequence is a noisy observation of an underlying hidden process represented as a first-order Markov chain. In other terms, long-range dependencies between observations are mediated via latent variables. It is important to note that the Markov property is only assumed for the hidden states and not for the observations themselves.

## 1.1 Model specification

HMM involves two interconnected models. The state model consists of a discrete-time, discrete-state, first order hidden Markov chain  $z_t \in \{1, \dots, K\}$  that transitions according to  $p(z_t|z_{t-1})$ . In turns, the observation model is governed by  $p(\mathbf{x}_t|z_t)$ , where  $\mathbf{x}_t$  are the observations, emissions or output. The corresponding joint distribution is

$$p(\mathbf{z}_{1:T}, \mathbf{x}_{1:T}) = p(\mathbf{z}_{1:T})p(\mathbf{x}_{1:T}|\mathbf{z}_{1:T}) = \left[ p(z_1) \prod_{t=2}^T p(z_t|z_{t-1}) \right] \left[ \prod_{t=1}^T p(\mathbf{x}_t|z_t) \right].$$

This is a specific instance of the state space model family in which the latent variables are discrete. Each single time slice corresponds to a mixture distribution with component densities given by  $p(\mathbf{x}_t|z_t)$ , thus HMM may be interpreted as an extension of a mixture model in which the choice of component for each observation is not selected independently but depends on the choice of component for the previous observation. In the case of a simple mixture model for an independent and identically distributed sample, the parameters of the transition matrix inside the  $i$ -th column are the same, so that the conditional distribution  $p(z_t|z_{t-1})$  is independent of  $z_{t-1}$ .

When the output is discrete, the observation model commonly takes the form of an observation matrix

$$p(\mathbf{x}_t = l | z_t = k, \theta) = B(k, l).$$

Alternatively, if the output is continuous, the observation model is frequently a conditional Gaussian

$$p(\mathbf{x}_t | z_t = k, \theta) = \mathcal{N}(\mathbf{x}_t | \mu_k, \Sigma_k).$$

The latter is equivalent to a Gaussian mixture model with cluster membership ruled by Markovian dynamics, also known as Markov Switching Models (MSM). In this context, multiple sequential observations tend to share the same location until they suddenly jump into a new cluster.

The non-stochastic quantities of the model are the observed sequence  $\mathbf{x}_t$ , the length of the observed sequence  $T$  and the number of hidden states  $K$ . The parameters of the models are  $\theta = (\pi_1, \theta_h, \theta_o)$ , where  $\pi_1$  is the initial state distribution,  $\theta_h$  are the parameters of the hidden model and  $\theta_o$  are the parameters of the state-conditional density function  $p(\mathbf{x}_t|z_t)$ . The form of  $\theta_h$  and  $\theta_o$  depends on the specification of each model. In the case under study, state transition is characterized by the  $K \times K$  sized transition matrix  $\mathbf{A} = \{a_{ij}\}$  with  $a_{ij} = p(z_t = j | z_{t-1} = i)$ .

The following Stan code illustrates the case of continuous observations where emissions are modeled as sampled from the Gaussian distribution with parameters  $\mu_k$  and  $\sigma_k$  for  $k \in \{1, \dots, K\}$ . Adaptation for categorical observations should follow the guidelines outlined in the manual (Stan Development Team 2017c, sec. 10.6).

```
data {
  int<lower=1> T;           // number of observations (length)
  int<lower=1> K;           // number of hidden states
  real x_t[T];             // observations
}

parameters {
```

```

// Discrete state model
simplex[K] p_1k;           // initial state probabilities
simplex[K] A_ij[K];        // transition probabilities
                           // A_ij[i][j] = p(z_t = j | z_{t-1} = i)

// Continuous observation model
ordered[K] mu_k;          // observation means
real<lower=0> sigma_k[K];  // observation standard deviations
}

```

## 1.2 The generative model

We write a routine in the R programming language for our generative model. Broadly speaking, this involves three steps:

1. The generation of parameters according to the priors  $\theta^{(0)} \sim p(\theta)$ .
2. The generation of the hidden path  $\mathbf{z}_{1:T}^{(0)}$  according to the transition model parameters.
3. The generation of the observed quantities based on the sampling distribution  $\mathbf{x}_t^{(0)} \sim p(\mathbf{x}_t | \mathbf{z}_{1:T}^{(0)}, \theta^{(0)})$ .

We break down the description of our code in these three steps.

```

runif_simplex <- function(T) {
  x <- -log(runif(T))
  x / sum(x)
}

hmm_generate <- function(K, T) {
  # 1. Parameters
  p_1k <- runif_simplex(K)
  A_ij <- t(replicate(K, runif_simplex(K)))
  mu_k <- sort(rnorm(K, 10 * 1:K, 1))
  sigma_k <- abs(rnorm(K))

  # 2. Hidden path
  z_t <- vector("numeric", T)

  z_t[1] <- sample(1:K, size = 1, prob = p_1k)
  for (t in 2:T)
    z_t[t] <- sample(1:K, size = 1, prob = A_ij[z_t[t - 1], ])

  # 3. Observations
  x_t <- vector("numeric", T)
  for (t in 1:T)
    x_t[t] <- rnorm(1, mu_k[z_t[t]], sigma_k[z_t[t]])

  list(x_t = x_t, z_t = z_t,
        theta = list(p_1k = p_1k, A_ij = A_ij,
                      mu_k = mu_k, sigma_k = sigma_k))
}

```

### 1.2.1 Generating parameters from the priors

The parameters to be generated include the  $K$ -sized initial state distribution vector  $\pi_1$  and the  $K \times K$  transition matrix  $\mathbf{A}$ . There are  $(K-1)(K+1)$  free parameters as the vector and each row of the matrix are simplexes.

We set up uniform priors for  $\pi_1$  and  $\mathbf{A}$ , a weakly informative Gaussian for the location parameter  $\mu_k$  and a weakly informative half-Gaussian that ensures positivity for the scale parameters  $\sigma_k$ . An ordinal constraint is imposed on the location parameter to restrict the exploration of the symmetric, degenerate mixture posterior surface to a single ordering of the parameters, thus solving the non-identifiability issues inherent to the model density (Betancourt 2017). In the simulation routine, the location parameters are adjusted to ensure that the observations are well-separated. We refer the reader to the [Prior Choice Recommendations](#) wiki article for very useful practical guidelines that are both computationally and statistically meaningful. Given the fixed quantity  $K$ , we draw one sample from the prior distributions  $\theta^{(0)} \sim p(\theta)$ .

### 1.2.2 Generating the hidden path

The initial hidden state is drawn from a multinomial distribution with one trial and event probabilities given by the initial state probability vector  $\pi_1^{(0)}$ . Given the fixed quantity  $T$ , the transition probabilities for each of the following steps  $t \in \{2, \dots, T\}$  are generated from a multinomial distribution with one trial and event probabilities given by the  $i$ -th row of the transition matrix  $\mathbf{A}_1^{(0)}$  where  $i$  is the state at the previous time step  $z_{t-1}^{(0)} = i$ . The hidden states are subsequently sampled based on these transition probabilities.

### 1.2.3 Generating data from the sampling distribution

The observation is drawn from a univariate Gaussian density with parameters  $\mu_k^{(0)}$  and  $\sigma_k^{(0)}$ .

## 1.3 Characteristics

One of the most powerful properties of HMM is the ability to exhibit some degree of invariance to local warping of the time axis. Allowing for compression or stretching of the time, the model accommodates for variations in speed. By specification of the latent model, the density function of the duration  $\tau$  in state  $i$  is given by

$$p_i(\tau) = (A_{ii})^\tau (1 - A_{ii}) \propto \exp(-\tau \ln A_{ii}),$$

which represents the probability that a sequence spends precisely  $\tau$  steps in state  $i$ . The expected duration conditional on starting in that state is

$$\bar{\tau}_i = \sum_{\tau=1}^{\infty} \tau p_i(\tau) = \frac{1}{1 - A_{ii}}.$$

The density is an exponentially decaying function of  $\tau$ , thus longer durations are more probable than shorter ones. In applications where this proves unrealistic, the diagonal coefficients of the transition matrix  $A_{ii} \forall i$  may be set to zero and each state  $i$  is explicitly associated with a probability distribution of possible duration times  $p(\tau|i)$  (Rabiner 1990).

## 1.4 Inference

There are several quantities of interest that can be inferred via different algorithms. Our code contains the implementation of the most relevant methods for unsupervised data: forward, forward-backward and Viterbi encoding algorithms. We acknowledge the authors of the Stan Manual for the thorough illustrations and code snippets, some of which served as a starting point for our own code. As estimation is treated later, we assume that model parameters  $\theta$  are known.

Table 1: Summary of the hidden quantities and their corresponding inference algorithm.

Name	Hidden Quantity	Availability at	Algorithm	Complexity
Filtering	$p(z_t \mathbf{x}_{1:t})$	$t$ (online)	Forward	$O(K^2T)$ $O(KT)$ if left-to-right
Smoothing	$p(z_t \mathbf{x}_{1:T})$	$T$ (offline)	Forward-backward	$O(K^2T)$ $O(KT)$ if left-to-right
Fixed lag smoothing	$p(z_{t-\ell} \mathbf{x}_{1:t}), \ell \geq 1$	$t + \ell$ (lagged)		
State prediction	$p(z_{t+h} \mathbf{x}_{1:t}), h \geq 1$	$t$		
Observation prediction	$p(x_{t+h} \mathbf{x}_{1:t}), h \geq 1$	$t$		
MAP Estimation	?	$T$	Viterbi encoding	$O(K^2T)$ $O(KT)$ if sparse
Probability of the evidence	$p(\mathbf{x}_{1:T})$	$T$	Forward	$O(K^2T)$ $O(KT)$ if left-to-right

### 1.4.1 Filtering

A filter infers the belief state at a given step  $t$  based on all the information available up to that point  $p(z_t|\mathbf{x}_{1:t})$ . It achieves better noise reduction than simply estimating the hidden state based on the current estimate  $p(z_t|\mathbf{x}_t)$ . The filtering process can be run online, or recursively, as new data streams in.

Filtered marginals can be computed recursively by means of the forward algorithm (Baum and Eagon 1967). Let  $\psi_t(j) = p(\mathbf{x}_t|z_t = j)$  be the local evidence at step  $t$  and  $\Psi(i, j) = p(z_t = j|z_{t-1} = i)$  be the transition probability. First, the one-step-ahead predictive density is computed

$$p(z_t = j|\mathbf{x}_{1:t-1}) = \sum_i \Psi(i, j)p(z_{t-1} = i|\mathbf{x}_{1:t-1}).$$

Acting as prior information, this quantity is updated with observed data at the step  $t$  using Bayes rule,

$$\begin{aligned} \alpha_t(j) &\triangleq p(z_t = j|\mathbf{x}_{1:t}) \\ &= p(z_t = j|\mathbf{x}_t, \mathbf{x}_{1:t-1}) \\ &= Z_t^{-1} \psi_t(j) p(z_t = j|\mathbf{x}_{1:t-1}) \end{aligned}$$

where the normalization constant is given by

$$Z_t \triangleq p(\mathbf{x}_t|\mathbf{x}_{1:t-1}) = \sum_{l=1}^K p(\mathbf{x}_t|z_t = l)p(z_t = l|\mathbf{x}_{1:t-1}) = \sum_{l=1}^K \psi_t(l)p(z_t = l|\mathbf{x}_{1:t-1}).$$

This predict-update cycle results in the filtered belief states at step  $t$ . As this algorithm only requires the evaluation of the quantities  $\psi_t(j)$  for each value of  $z_t$  for every  $t$  and fixed  $\mathbf{x}_t$ , the posterior distribution of the latent states is independent of the form of the observation density or indeed of whether the observed variables are continuous or discrete (Jordan 2003).

Let  $\alpha_t$  be a  $K$ -sized vector with the filtered belief states at step  $t$ ,  $\psi_t(j)$  be the  $K$ -sized vector of local evidence at step  $t$ ,  $\Psi$  be the transition matrix and  $\mathbf{u} \odot \mathbf{v}$  be the Hadamard product, representing element-wise vector multiplication. Then, the Bayesian updating procedure can be expressed in matrix notation as

$$\alpha_t \propto \psi_t \odot (\Psi^T \alpha_{t-1}).$$

In addition to computing the hidden states, the algorithm yields the log probability of the evidence

$$\log p(\mathbf{x}_{1:T}|\theta) = \sum_{t=1}^T \log p(\mathbf{x}_t|\mathbf{x}_{1:t-1}) = \sum_{t=1}^T \log Z_t.$$

```
transformed parameters {
  vector[K] unalpha_tk[T];

  { // Forward algorithm log p(z_t = j | x_{1:t})
    real accumulator[K];

    unalpha_tk[1] = log(p_1k) + normal_lpdf(x_t[1] | mu_k, sigma_k);

    for (t in 2:T) {
      for (j in 1:K) { // j = current (t)
        for (i in 1:K) { // i = previous (t-1)
          // Murphy (2012) p. 609 eq. 17.48
          // belief state      + transition prob + local evidence at t
          accumulator[i] = unalpha_tk[t-1, i]
            + log(A_ij[i, j])
            + normal_lpdf(x_t[t] | mu_k[j], sigma_k[j]);
        }
        unalpha_tk[t, j] = log_sum_exp(accumulator);
      }
    }
  } // Forward
}
```

The Stan code makes evident that the time complexity of the algorithm is  $O(K^2T)$ : there are  $K \times K$  iterations within each of the  $T$  iterations of the outer loop. Brute-forcing through all possible hidden states  $K^T$  would prove prohibitive for realistic problems as time complexity increases exponentially with sequence length  $O(K^T T)$ .

The implementation is representative of the matrix notation in Murphy (2012 eq. 17.48). The `accumulator` variable carries the element-wise operations for all possible previous states which are later combined as indicated by the matrix multiplication.

Since log domain should be preferred to avoid numerical underflow, multiplications are translated into sums of logs. Furthermore, we use Stan’s implementation of the linear sums on the log scale to prevent underflow and overflow in the exponentiation (Stan Development Team 2017c, 141). In consequence, `unalpha_tk` represents the forward quantity in log scale and needs to be exponentially normalized for interpretability.

```
generated quantities {
  vector[K] alpha_tk[T];

  { // Forward algorithm
    for (t in 1:T)
      alpha_tk[t] = softmax(unalpha_tk[t]);
  } // Forward
}
```

The unnormalized forward probability is a sufficient statistics that will be used for estimation in the model and, as such, should be part of either the `model` or the `transformed parameters` blocks. We chose the latter to keep track of the estimates. We expand on estimation afterward.

### 1.4.2 Smoothing

A smoother infers the belief state at a given state based on all the observations or evidence  $p(z_t | \mathbf{x}_{1:T})$ . Although noise and uncertainty are significantly reduced as a result of conditioning on past and future data, the smoothing process can only be run offline.

Inference can be done by means of the forward-backward algorithm, also known as the Baum-Welch algorithm (Baum and Eagon 1967; Baum et al. 1970). Let  $\gamma_t(j)$  be the desired smoothed posterior marginal,

$$\gamma_t(j) \triangleq p(z_t = j | \mathbf{x}_{1:T}),$$

$\alpha_t(j)$  be the filtered belief state at the step  $t$  as defined previously, and  $\beta_t(j)$  be the conditional likelihood of future evidence given that the hidden state at step  $t$  is  $j$ ,

$$\beta_t(j) \triangleq p(\mathbf{x}_{t+1:T} | z_t = j).$$

Then, the chain of smoothed marginals can be segregated into the past and the future components by conditioning on the belief state  $z_t$ ,

$$\gamma_t(j) = p(z_t = j | \mathbf{x}_{1:T}) \propto p(z_t = j, \mathbf{x}_{t+1:T} | \mathbf{x}_{1:t}) \propto p(z_t = j | \mathbf{x}_{1:t}) p(\mathbf{x}_{t+1:T} | z_t = j) \propto \alpha_t(j) \beta_t(j).$$

The future component can be computed recursively from right to left:

$$\begin{aligned} \beta_{t-1}(i) &= p(\mathbf{x}_{t:T} | z_{t-1} = i) \\ &= \sum_{j=1}^K p(z_t = j, \mathbf{x}_t, \mathbf{x}_{t+1:T} | z_{t-1} = i) \\ &= \sum_{j=1}^K p(\mathbf{x}_{t+1:T} | z_t = j) p(z_t = j, \mathbf{x}_t | z_{t-1} = i) \\ &= \sum_{j=1}^K p(\mathbf{x}_{t+1:T} | z_t = j) p(\mathbf{x}_t | z_t = j) p(z_t = j | z_{t-1} = i) \\ &= \sum_{j=1}^K \beta_t(j) \psi_t(j) \Psi(i, j) \end{aligned}$$

Let  $\beta_t$  be a  $K$ -sized vector with the conditional likelihood of future evidence given the hidden state at step  $t$ . Then, the backward procedure can be expressed in matrix notation as

$$\beta_{t-1} \propto \Psi(\psi_t \odot \beta_t).$$

At the last step, the base case is given by

$$\beta_T(i) = p(\mathbf{x}_{T+1:T} | z_T = i) = p(\emptyset | z_T = i) = 1.$$

Intuitively, the forward-backward algorithm passes information from left to right and then from right to left, combining them at each node. A straightforward implementation of the algorithm runs in  $O(K^2T)$  time because of the  $K \times K$  matrix multiplication at each step. There is a significant reduction if the transition matrix is sparse, for example a left-to-right transition matrix runs in  $O(TK)$  time. Additional assumptions about the form of the transition matrix may ease complexity further, for example reducing the time to  $O(TK \log K)$  if  $\psi(i, j) \propto \exp(-\sigma^2 | \mathbf{z}_i - \mathbf{z}_j |)$ .

```

functions {
  vector normalize(vector x) {
    return x / sum(x);
  }
}

generated quantities {
  vector[K] unbeta_tk[T];
  vector[K] ungamma_tk[T];

  vector[K] beta_tk[T];
  vector[K] gamma_tk[T];

  { // Forward algortihm
    for (t in 1:T)
      alpha_tk[t] = softmax(unalpha_tk[t]);
  } // Forward

  { // Backward algorithm log p(x_{t+1:T} | z_t = j)
    real accumulator[K];

    for (j in 1:K)
      unbeta_tk[T, j] = 1;

    for (tforward in 0:(T-2)) {
      int t;
      t = T - tforward;

      for (j in 1:K) { // j = previous (t-1)
        for (i in 1:K) { // i = next (t)
          // Murphy (2012) Eq. 17.58
          // backwards t + transition prob + local evidence at t
          accumulator[i] = unbeta_tk[t, i]
            + log(A_ij[j, i])
            + normal_lpdf(x_t[t] | mu_k[i], sigma_k[i]);
        }
        unbeta_tk[t-1, j] = log_sum_exp(accumulator);
      }
    }

    for (t in 1:T)
      beta_tk[t] = softmax(unbeta_tk[t]);
  } // Backward

  { // forward-backward algorithm log p(z_t = j | x_{1:T})
    for(t in 1:T) {
      ungamma_tk[t] = alpha_tk[t] .* beta_tk[t];
    }

    for(t in 1:T)
      gamma_tk[t] = normalize(ungamma_tk[t]);
  } // forward-backward
}

```



The reader should not be deceived by the similarity to the code shown in the filtering section. Note that the indices in the log transition matrix are inverted and the evidence is now computed for the next state. We need to invert the time index as backward ranges are available in Stan.

### 1.4.3 MAP: Viterbi

It is also of interest to compute the most probable state sequence or path,

$$\mathbf{z}^* = \operatorname{argmax}_{\mathbf{z}_{1:T}} p(\mathbf{z}_{1:T} | \mathbf{x}_{1:T}).$$

The jointly most probable sequence of states can be inferred by means of maximum a posterior (MAP) estimation. It is not necessarily the same as the sequence of marginally most probable states given by the maximizer of the posterior marginals (MPM),

$$\hat{\mathbf{z}} = (\operatorname{argmax}_{z_1} p(z_1 | \mathbf{x}_{1:T}), \dots, \operatorname{argmax}_{z_T} p(z_T | \mathbf{x}_{1:T})),$$

which maximizes the expected number of correct individual states.

The MAP estimate is always globally consistent: while locally a state may be most probable at a given step, the Viterbi or max-sum algorithm decodes the most likely single plausible path (Viterbi 1967). Furthermore, the MPM sequence may have zero joint probability if it includes two successive states that, while being individually the most probable, are connected in the transition matrix by a zero. On the other hand, MPM can be considered more robust since the state at each step is estimated by averaging over its neighbors rather than conditioning on a specific value of them.

The Viterbi algorithm is an adaptation of the forward-backward algorithm where the forward pass becomes a max-product and the backwards pass relies on a traceback procedure to recover the most probable path. In simple terms, once the most probable state  $z_t$  is estimated, the procedure conditions the previous states on it. Let  $\delta_t(j)$  be the probability of arriving to the state  $j$  at step  $t$  given the most probable path was taken,

$$\delta_t(j) \triangleq \max_{z_1, \dots, z_{t-1}} p(\mathbf{z}_{1:t-1}, z_t = j | \mathbf{x}_{1:t}).$$

The most probable path to state  $j$  at step  $t$  consists of the most probable path to some other state  $i$  at point  $t-1$ , followed by a transition from  $i$  to  $j$ ,

$$\delta_t(j) = \max_i \delta_{t-1}(i) \psi(i, j) \psi_t(j).$$

Additionally, the most likely previous state on the most probable path to  $j$  at step  $t$  is given by

$$a_t(j) = \operatorname{argmax}_i \delta_{t-1}(i) \psi(i, j) \psi_t(j).$$

By initializing with  $\delta_1 = \pi_j \phi_1(j)$  and terminating with the most probable final state  $z_T^* = \operatorname{argmax}_i \delta_T(i)$ , the most probable sequence of states is estimated using the traceback,

$$z_t^* = a_{t+1}(z_{t+1}^*).$$

It is advisable to work in the log domain to avoid numerical underflow,

$$\delta_t(j) \triangleq \max_{\mathbf{z}_{1:t-1}} \log p(\mathbf{z}_{1:t-1}, z_t = j | \mathbf{x}_{1:t}) = \max_i \log \delta_{t-1}(i) + \log \psi(i, j) + \log \psi_t(j).$$

As with the backward-forward algorithm, the time complexity of Viterbi is  $O(K^2T)$  and the space complexity is  $O(KT)$ . If the transition matrix has the form  $\psi(i, j) \propto \exp(-\sigma^2 \|\mathbf{z}_i - \mathbf{z}_j\|^2)$ , implementation runs in  $O(TK)$  time.

```
generated quantities {
  int<lower=1, upper=K> zstar_t[T];
  real logp_zstar_t;

  { // Viterbi algorithm
    int a_tk[T, K];          // backpointer to the most likely previous state
                                // on the most probable path
    real delta_tk[T, K];     // max prob for the sequence up to t
                                // that ends with an emission from state k

    for (j in 1:K)
      delta_tk[1, K] = normal_lpdf(x_t[1] | mu_k[j], sigma_k[j]);

    for (t in 2:T) {
      for (j in 1:K) { // j = current (t)
        delta_tk[t, j] = negative_infinity();
        for (i in 1:K) { // i = previous (t-1)
          real logp;
          logp = delta_tk[t-1, i]
                + log(A_ij[i, j])
                + normal_lpdf(x_t[t] | mu_k[j], sigma_k[j]);
          if (logp > delta_tk[t, j]) {
            a_tk[t, j] = i;
            delta_tk[t, j] = logp;
          }
        }
      }
    }

    logp_zstar_t = max(delta_tk[T]);

    for (j in 1:K)
      if (delta_tk[T, j] == logp_zstar_t)
        zstar_t[T] = j;

    for (t in 1:(T - 1)) {
      zstar_t[T - t] = a_tk[T - t + 1, zstar_t[T - t + 1]];
    }
  }
}
```

The variable `delta_tk` is a straightforward implementation of the corresponding equation. `a_tk` is the traceback needed to compute the most probable sequence of states after the most probably final state `zstar_t` is computed.

## 1.5 Parameter estimation

The probability of a state at a certain time step  $t$ , given the history of evidence  $\mathbf{x}_{1:t}$ , is an average over all possible previous states (in other terms, previous state are marginalized out). The recurrent nature of the forward probability implies that the last iteration contains all the sufficient information to estimate the

parameters.

```
model {  
  target += log_sum_exp(unalpha_tk[T]); // Note: update based only on last unalpha_tk  
}
```

A sufficient statistic captures all the information that the sample contains about the parameter vector, achieving the maximal amount of data reduction possible without loss of information. More formally, a statistic  $T(\mathbf{x}_t)$  is a sufficient statistic for the parameter vector  $\theta$  if the conditional distribution of the sample  $\mathbf{x}_t$  given the value of  $T(\mathbf{x}_t)$  does not depend on  $\theta$  (Casella and Berger 2002).

As illustrated in the Manual (Stan Development Team 2017c, 353), sufficient statistics may be exploited to generate large efficiency gains: some models can be restated in an alternative, less complex way to produce a likelihood that is proportional to the expanded model.

```
hmm_init <- function(K, x_t) {  
  clasif <- kmeans(x_t, K)  
  init.mu <- by(x_t, clasif$cluster, mean)  
  init.sigma <- by(x_t, clasif$cluster, sd)  
  init.order <- order(init.mu)  
  
  list(  
    mu_k = init.mu[init.order],  
    sigma_k = init.sigma[init.order]  
  )  
}  
  
hmm_fit <- function(K, x_t) {  
  rstan_options(auto_write = TRUE)  
  options(mc.cores = parallel::detectCores())  
  
  stan.model = 'stan/hmm_gaussian.stan'  
  stan.data = list(  
    T = length(x_t),  
    K = K,  
    x_t = x_t  
  )  
  
  stan(file = stan.model,  
    data = stan.data, verbose = T,  
    iter = 400, warmup = 200,  
    thin = 1, chains = 1,  
    cores = 1, seed = 900,  
    init = function(){hmm_init(K, x_t)})  
}
```

As we expect high multimodality in the posterior density, we use a clustering algorithm to feed the sampler with initialization values for the location and scale parameters. Although K-means is not truthful to the real model because it does not consider the time-dependent nature of the data, it provides an educated guess.


## 1.6 A walk-through

We draw one sample of length  $T = 500$  from a data generating process with  $K = 3$  latent states. We fit the model using the Stan code and the initialization methodology introduced previously.

The estimates are extremely efficient as expected when dealing with generated data. The Markov Chain

are well behaved as diagnosed by the low Monte Carlo standard error, the high effective sample size and the near-one shrink factor of Gelman and Rubin (1992). Although not shown, further diagnostics confirm satisfactory mixing, convergence and the absence of divergences. Point estimates and credibility intervals are provided by rstan's `summary` function.

We extract the samples for some quantities of interest, namely the filtered probabilities vector  $\alpha_t$ , the smoothed probability vector  $\gamma_t$  and the most probable hidden path  $\mathbf{z}^*$ . As an informal assessment that our software recover the hidden states correctly, we observe that the filtered probability, the smoothed probability and the most likely path are all reasonable accurate to the true values.



We identify an insignificant quantity of misclassifications product of the stochastic nature of our software.

As expected, the MAP estimate recovers the simulated hidden path with no label switching due because of the ordinal constraints.





Table 2: Hard classification according to the Viterbi Algorithm.

	$\hat{z}_t = 1$	$\hat{z}_t = 2$	$\hat{z}_t = 2$
$z_t = 1$	154	1	0
$z_t = 2$	4	227	2
$z_t = 3$	0	5	107

Finally, we plot the observed series colored according to the jointly most likely state.



## 2 The Input-Output Hidden Markov Model

The IOHMM is an architecture proposed by Bengio and Frasconi (1995) to map input sequences, sometimes called the control signal, to output sequences. It is a probabilistic framework that can deal with general sequence processing tasks such as production, classification and prediction. The main difference with Hidden Markov Models (HMM), which are part of the unsupervised learning paradigm, is the capability to learn the output sequence itself instead of the distribution of the output sequence.

### 2.1 Definitions

As with HMM, IOHMM involves two interconnected models,

$$\begin{aligned} z_t &= f(z_{t-1}, \mathbf{u}_t) \\ \mathbf{x}_t &= g(z_t, \mathbf{u}_t). \end{aligned}$$

The first line corresponds to the state model, which consists of discrete-time, discrete-state hidden states  $z_t \in \{1, \dots, K\}$  whose transition depends on the previous hidden state  $z_{t-1}$  and the input vector  $\mathbf{u}_t \in \mathbb{R}^M$ . Additionally, the observation model is governed by  $g(z_t, \mathbf{u}_t)$ , where  $\mathbf{x}_t \in \mathbb{R}^R$  is the vector of observations, emissions or output. The corresponding joint distribution is

$$p(\mathbf{z}_{1:T}, \mathbf{x}_{1:T} | \mathbf{u}_t).$$

In the proposed parametrization with continuous inputs and outputs, the state model involves a multinomial regression whose parameters depend on the previous state taking the value  $i$ ,

$$p(z_t | \mathbf{x}_t, \mathbf{u}_t, z_{t-1} = i) = \text{softmax}^{-1}(\mathbf{u}_t \mathbf{w}_i),$$

and the observation model is built upon a linear regression with Gaussian error and parameters depending on the current state taking the value  $j$ ,

$$p(\mathbf{x}_t | \mathbf{u}_t, z_t = j) = \mathcal{N}(\mathbf{u}_t \mathbf{b}_j, \Sigma_j)$$

IOHMM adapts the logic of HMM to allow for input and output vectors, retaining its fully probabilistic quality. Hidden states are assumed to follow a multinomial distribution that depends on the input sequence. The transition probabilities  $\Psi_t(i, j) = p(z_t = j | z_{t-1} = i, \mathbf{u}_t)$ , which govern the state dynamics, are driven by the control signal as well.

As for the output sequence, the local evidence at time  $t$  now becomes  $\psi_t(j) = p(\mathbf{x}_t | z_t = j, \eta_t)$ , where  $\eta_t = E \langle \mathbf{x}_t | z_t, \mathbf{u}_t \rangle$  can be interpreted as the expected location parameter for the probability distribution of the emission  $\mathbf{x}_t$  conditional on the input vector  $\mathbf{u}_t$  and the hidden state  $z_t$ .

The actual form of the emission density  $p(\mathbf{x}_t, \eta_t)$  can be discrete or continuous. In case of sequence classification or symbolic mutually exclusive emissions, it is possible to set up the multinomial distribution by running the softmax function over the estimated outputs of all possible states. In this case, we approximate continuous observations with the Gaussian density, the target is estimated as a linear combination of these outputs.

The adaptation of the data and parameters blocks is straightforward: we add the number of input variables  $M$ , the array of input vectors  $\mathbf{u\_t}$ , the regressors  $\mathbf{b\_km}$  and the residual standard deviation  $\mathbf{s\_k}$ .

```
data {
  int<lower=1> T;           // number of observations (length)
  int<lower=1> K;           // number of hidden states
  int<lower=1> M;           // size of the input vector

  real x_t[T];             // output (scalar so far)
  vector[M] u_tm[T];       // input vectors
}

parameters {
  // Discrete state model
  simplex[K] p_1k;         // initial state probabilities
  vector[M] w_km[K];       // state regressors

  // Continuous observation model
  vector[M] b_km[K];       // mean regressors
  real<lower=0.0001> s_k[K]; // residual standard deviations
}
```

## 2.2 Inference

### 2.2.1 Filtering

The filtered marginals are computed recursively by adjusting the forward algorithm to consider the input sequence,

$$\begin{aligned}
\alpha_t(j) &\triangleq p(z_t = j | \mathbf{x}_{1:t}, \mathbf{u}_{1:t}) \\
&= \sum_{i=1}^K p(z_t = j | z_{t-1} = i, \mathbf{x}_t, \mathbf{u}_t) p(z_{t-1} = i | \mathbf{x}_{1:t-1}, \mathbf{u}_{1:t-1}) \\
&= \sum_{i=1}^K p(\mathbf{x}_t | z_t = j, \mathbf{u}_t) p(z_t = j | z_{t-1} = i, \mathbf{u}_t) p(z_{t-1} = i | \mathbf{x}_{1:t-1}, \mathbf{u}_{1:t-1}) \\
&= \psi_t(j) \sum_{i=1}^K \Psi_t(i, j) \alpha_{t-1}(i).
\end{aligned}$$

The implementation in Stan requires one modification: the time-dependent transition probability matrix is now computed as the linear combination of the input variables and the parameters of the multinomial regression that drives the latent process.

```

transformed parameters {
  vector[K] unalpha_tk[T];

  vector[K] unA_ij[T];
  vector[K] A_ij[T];

  vector[K] oblik_tk[T];

  { // Transition probability matrix p(z_t = j | z_{t-1} = i, u_t)
    unA_ij[1] = p_1k; // Filler
    A_ij[1] = p_1k; // Filler
    for (t in 2:T) {
      for (j in 1:K) { // j = current (t)
        unA_ij[t][j] = u_t[t]' * w_km[j];
      }
      A_ij[t] = softmax(unA_ij[t]);
    }
  }

  { // Evidence (observation likelihood)
    for(t in 1:T) {
      for(j in 1:K) {
        oblik_tk[t][j] = normal_lpdf(x_t[t] | mu_k[j], s_k[j]);
      }
    }
  }

  { // Forward algorithm log p(z_t = j | x_{1:t})
    real accumulator[K];

    for(j in 1:K)
      unalpha_tk[1][j] = log(p_1k[j]) + oblik_tk[1][j];

    for (t in 2:T) {
      for (j in 1:K) { // j = current (t)
        for (i in 1:K) { // i = previous (t-1)
          // Murphy (2012) Eq. 17.48
          // belief state + transition prob + local evidence at t

```



```

        accumulator[i] = unalpha_tk[t-1, i]
                        + log(A_ij[t][i])
                        + oblik_tk[t][j];
    }
    unalpha_tk[t, j] = log_sum_exp(accumulator);
}
}
} // Forward
}

```

### 2.2.2 Smoothing

A smoother infers the belief state at a given step based on all the observations or evidence,

$$\begin{aligned}\gamma_t(j) &\triangleq p(z_t = j | \mathbf{x}_{1:T}, \mathbf{u}_{1:T}) \\ &\propto \alpha_t(j) \beta_t(j),\end{aligned}$$

where

$$\beta_{t-1}(i) \triangleq p(\mathbf{x}_{t:T} | z_{t-1} = i, \mathbf{u}_{t:T}).$$

Similarly, inference about the smoothed posterior marginal requires the adaptation of the forward-backward algorithm to consider the input sequence in both components  $\alpha_t(j)$  and  $\beta_t(j)$ . The latter now becomes

$$\begin{aligned}\beta_{t-1}(i) &\triangleq p(\mathbf{x}_{t:T} | z_{t-1} = i, \mathbf{u}_{t:T}) \\ &= \sum_{j=1}^K \psi_t(j) \Psi_t(i, j) \beta_t(j).\end{aligned}$$

Once adjusted the transition probability matrix, the Stan code for the forward-backward algorithm need no further modification.

### 2.2.3 MAP: Viterbi

The Stan code for the Viterbi encoding algorithm need no further modification.

## 2.3 Parameter estimation

The parameters of the models are  $\theta = (\pi_1, \theta_h, \theta_o)$ , where  $\pi_1$  is the initial state distribution,  $\theta_h$  are the parameters of the hidden model and  $\theta_o$  are the parameters of the state-conditional density function  $p(\mathbf{x}_t | z_t = j, \mathbf{u}_t)$ . State transition is characterized by a multinomial regression with parameters  $\mathbf{w}_k$  for  $k \in \{1, \dots, K\}$ , while emissions are modeled by a linear regression with Gaussian error and parameters  $\mathbf{b}_k$  and  $\Sigma_k$  for  $k \in \{1, \dots, K\}$ .

Estimation can be run under both the maximum likelihood and Bayesian frameworks. Although it is a straightforward procedure when the data is fully observed, in practice the latent states  $\mathbf{z}_{1:T}$  are hidden. The most common approach is the application of the EM algorithm to find either the maximum likelihood or the

maximum a posteriori estimates. Bengio and Frasconi (1995) proposes a straightforward modification of the EM algorithm. The application of sigmoidal functions, for example the logistic or softmax transforms for the hidden transition model, requires numeric optimization via gradient ascent or similar methods for the M step. In this work, we exploit Stan’s capabilities to produce a sampler that explores the posterior density of the model parameters.

## 2.4 A walk-through the IOHMM

### 2.4.1 Numerical stability for the softmax function

A digression! The softmax function, or normalized exponential function, can suffer from over or underflow in the exponentials. A naive implementation may fail:

```
x <- 10^(1:5)
exp(x) / sum(exp(x))
```

```
## [1] 0 0 NaN NaN NaN
```

A well known, safer implementation exploits the fact that softmax is location invariant, ie  $\text{softmax}(\mathbf{x}) = \text{softmax}(\mathbf{x} + c)$  for any constant  $c$ . Subtracting the maximum value produces a new vector with non-positive entries, ruling out overflows, and at least one zero element, guaranteeing at least one significant term in the denominator.

```
logsumexp <- function(x) {
  y = max(x)
  y + log(sum(exp(x - y)))
}
```

```
softmax <- function(x) {
  exp(x - logsumexp(x))
}
```

```
softmax(x)
```

```
## [1] 0 0 0 0 1
```

This is already taken care in Stan (Stan Development Team 2017c, 474).

### 2.4.2 Back to the walk-through

We first adapt the R routine for our new generative model. The arguments are the sequence length  $T$ , the number of discrete hidden states  $K$ , the input matrix  $\mathbf{u}$ , the initial state distribution vector  $\pi_1$ , a matrix with the parameters of the multinomial regression that rules the hidden states dynamics  $\mathbf{w}$ , the name of a function drawing samples from the observation distribution and its arguments.

The initial hidden state is drawn from a multinomial distribution with one trial and event probabilities given by the initial state probability vector  $\pi_1$ . The transition probabilities for each of the following steps  $t \in \{2, \dots, T\}$  are generated from a multinomial regression with vector parameters  $\mathbf{w}_k$ , one set per possible hidden state  $k \in \{1, \dots, K\}$ , and covariates  $\mathbf{u}_t$ . The hidden states are subsequently sampled based on these transition probabilities.

The observation at each step may generate from a Gaussian with parameters  $\mu_k$  and  $\sigma_k$ , one set per possible hidden state.

```
iohmm_generate <- function(T) {
  # 1. Parameters
```

```

K <- 3
M <- 4
u_tm <- matrix(rnorm(T * M), nrow = T, ncol = M, byrow = TRUE)
w_km <- matrix(
  c(1.2, 0.5, 0.3, 0.1, 0.5, 1.2, 0.3, 0.1, 0.5, 0.1, 1.2, 0.1),
  nrow = K, ncol = M, byrow = TRUE)
b_km <- matrix(
  c(5.0, 6.0, 7.0, 0.5, 1.0, 5.0, 0.1, -0.5, 0.1, -1.0, -5.0, 0.2),
  nrow = K, ncol = M, byrow = TRUE)
sigma_k <- c(0.2, 1.0, 2.5)
p_1k <- c(0.4, 0.2, 0.4)

p.mat <- matrix(0, nrow = T, ncol = K)
p.mat[1, ] <- p_1k

# 2. Hidden path
z_t <- vector("numeric", T)
z_t[1] <- sample(x = 1:K, size = 1, replace = FALSE, prob = p_1k)
for (t in 2:T) {
  p.mat[t, ] <- softmax(sapply(1:K, function(j) {u_tm[t, ] %*% w_km[j, ]}))
  z_t[t] <- sample(x = 1:K, size = 1, replace = FALSE, prob = p.mat[t, ])
}

# 3. Observations
x_t <- vector("numeric", T)
for (t in 1:T) {
  x_t[t] <- rnorm(1, u_tm[t, ] %*% b_km[z_t[t], ], sigma_k[z_t[t]])
}

list(
  u_tm = u_tm,
  z_t = z_t,
  x_t = x_t,
  theta = list(p_1k = p_1k, w_km = w_km,
               b_km = b_km, sigma_k = sigma_k, p.mat = p.mat)
)
}


```

We draw one sample of length  $T = 500$  from a data generating process with  $K = 3$  latent states and run an exploratory analysis of the observed quantities.


```

set.seed(900)
K <- 3
T_length <- 500
dataset <- iohmm_generate(T_length)

```



We observe how the chosen values for the parameters affect the generated data. For example, the relationship between the third input  $\mathbf{u}_3$  and the output  $\mathbf{x}_t$  is positive, indifferent and negative for the hidden states  $K = 1, 2, 3$  respectively. The true slopes are 7, 0.1 and -5.



We then analyse the relationship between the input and the state probabilities, which are usually hidden in applications with real data. The pairs  $\{\mathbf{u}_1, p(z_t = 1)\}$ ,  $\{\mathbf{u}_2, p(z_t = 2)\}$  and  $\{\mathbf{u}_3, p(z_t = 3)\}$  show the strongest relationships because of values of true regression parameters: those inputs take the largest weight in each state, namely  $w_{11} = 1.2$ ,  $w_{22} = 1.2$  and  $w_{33} = 1.2$ .

We run the software to draw samples from the posterior density of model parameters and other hidden quantities.

We rely on several diagnostic statistics and plots provided by rstan (Stan Development Team 2017a) and shinystan (Stan Development Team 2017b) to assess mixing, convergence and the absence of divergences. Label switching hinders the comparison between true and observed parameters.

Table 3: Estimated parameters and hidden quantities. *MCSE* = Monte Carlo Standard Error, *SE* = Standard Error, *ESS* = Effective Sample Size.

	True	Mean	MCSE	SE	$q_{10\%}$	$q_{50\%}$	$q_{90\%}$	ESS	$\hat{R}$
p_1k[1]	0.4	0.24	0.01	0.18	0.04	0.20	0.50	200.00	1.00

	True	Mean	MCSE	SE	$q_{10\%}$	$q_{50\%}$	$q_{90\%}$	ESS	$\hat{R}$
p_1k[2]	0.2	0.26	0.01	0.19	0.05	0.23	0.54	200.00	1.00
p_1k[3]	0.4	0.50	0.02	0.22	0.20	0.49	0.79	200.00	1.00
w_km[1,1]	1.2	-0.06	0.25	2.71	-3.74	0.46	3.13	121.58	1.00
w_km[1,2]	0.5	-0.05	0.21	2.87	-3.55	-0.08	3.82	189.90	1.02
w_km[1,3]	0.3	-0.07	0.31	2.92	-3.75	-0.26	3.69	91.49	1.00
w_km[1,4]	0.1	0.44	0.23	2.61	-3.14	0.37	3.55	128.73	1.00
w_km[2,1]	0.5	-0.06	0.25	2.72	-3.73	0.48	3.13	120.68	1.00
w_km[2,2]	1.2	-0.13	0.21	2.85	-3.75	-0.22	3.73	189.58	1.01
w_km[2,3]	0.3	-0.07	0.31	2.92	-3.80	-0.26	3.60	90.67	1.00
w_km[2,4]	0.1	0.58	0.23	2.60	-3.02	0.61	3.74	129.67	1.00
w_km[3,1]	0.5	-0.25	0.25	2.73	-4.01	0.33	3.01	120.45	1.00
w_km[3,2]	0.1	-0.09	0.21	2.86	-3.62	-0.15	3.72	189.68	1.01
w_km[3,3]	1.2	-0.09	0.31	2.93	-3.80	-0.28	3.62	90.45	1.00
w_km[3,4]	0.1	0.48	0.23	2.59	-3.07	0.46	3.64	130.86	1.00
b_km[1,1]	5.0	-0.04	0.02	0.24	-0.35	-0.03	0.25	200.00	1.00
b_km[1,2]	1.0	-1.11	0.01	0.18	-1.33	-1.11	-0.88	200.00	1.00
b_km[1,3]	0.1	-4.91	0.01	0.19	-5.12	-4.93	-4.66	200.00	1.00
b_km[1,4]	6.0	0.15	0.02	0.21	-0.10	0.14	0.42	200.00	1.00
b_km[2,1]	5.0	4.98	0.00	0.02	4.97	4.98	5.00	200.00	1.00
b_km[2,2]	-1.0	5.99	0.00	0.02	5.97	5.99	6.01	200.00	1.00
b_km[2,3]	7.0	7.00	0.00	0.02	6.98	7.00	7.02	200.00	1.00
b_km[2,4]	0.1	0.49	0.00	0.02	0.47	0.49	0.51	200.00	1.00
b_km[3,1]	-5.0	0.98	0.01	0.09	0.88	0.98	1.11	200.00	1.00
b_km[3,2]	0.5	4.99	0.01	0.09	4.88	4.99	5.11	200.00	1.01
b_km[3,3]	-0.5	0.10	0.01	0.09	-0.02	0.11	0.21	173.97	1.00
b_km[3,4]	0.2	-0.59	0.01	0.10	-0.73	-0.60	-0.46	200.00	1.01
sigma_k[1]	0.2	2.61	0.01	0.17	2.40	2.60	2.83	200.00	0.99
sigma_k[2]	1.0	0.20	0.00	0.01	0.18	0.20	0.22	200.00	1.00
sigma_k[3]	2.5	1.07	0.01	0.07	0.98	1.07	1.17	200.00	1.00


While mixing and convergence is extremely efficient, as expected when dealing with generated data, we note that the regression parameters for the latent states are the worst performers. The smaller effective size translates into higher Monte Carlo standard error and broader credibility intervals. One possible reason is that softmax is invariant to change in location, thus the parameters of a multinomial regression do not have a natural center and become harder to estimate.

We assess that our software recover the hidden states correctly. Due to label switching, the states generated under the labels 1 through 3 were recovered in a different order. In consequence, we decide to relabel the observations based on the best fit. This would not pose an inconvenient with real data as the hidden states are never observed.

Table 4: Simulated hidden states are relabeled to match the order of the recovered states.

	$z_1$	$z_2$	$z_3$
$z_1^r$	0	0	183
$z_2^r$	152	0	0
$z_3^r$	0	165	0

The filtered probability is overly accurate, as expected, making the additional information contained in the smoothed seem of little value.



The confusion matrix makes evident that, under ideal conditions, the sampler works as intended.

Table 5: Hard classification according to the filtered probability.

	$\hat{z}_t = 1$	$\hat{z}_t = 2$	$\hat{z}_t = 2$
$z_t^r = 1$	155	1	6
$z_t^r = 2$	6	151	7
$z_t^r = 3$	22	0	152

Similarly, the Viterbi algorithm recovers the expected most probably hidden state.




Table 6: Hard classification according to the Viterbi Algorithm.

	$\hat{z}_t = 1$	$\hat{z}_t = 2$	$\hat{z}_t = 2$
$z_t^r = 1$	0	151	1
$z_t^r = 2$	4	7	154
$z_t^r = 3$	155	6	22

### 3 Acknowledgements

The members of the Stan Development Team for being so passionate about Bayesian statistics and technically unbeatable. Special mentions to Aaron Goodman, Ben Bales and Bob Carpenter for their active participation in the discussions held in Stan forums for [HMM with constraints](#) and [HMMM](#). Although not strictly related, the discussion remains very valuable for the current Case Study.

This Case Study builds on top of my GSoC Project: [Bayesian Hierarchical Hidden Markov Models applied to financial time series](#). I thank Brian Peterson for being a great project leader and Michael Weylandt for being awesomely Bayesian and contributing with great statistical insight.

### 4 Original Computing Environment

```
## Warning in readLines(file.path(Sys.getenv("HOME"), ".R/Makevars")):
## incomplete final line found on 'C:\Users\Bebop\.R/Makevars'
## CXXFLAGS=-O3 -Wno-unused-variable -Wno-unused-function
```



```
## Session info -----
## setting value
## version R version 3.4.1 (2017-06-30)
## system x86_64, mingw32
## ui RTerm
## language (EN)
## collate Spanish_Argentina.1252
## tz America/Buenos_Aires
## date 2017-12-16

## Packages -----

## package      * version  date      source
## BH            1.65.0-1 2017-08-24 CRAN (R 3.4.1)
## colorspace    1.3-2     2016-12-14 CRAN (R 3.4.1)
## dichromat     2.0-0     2013-01-24 CRAN (R 3.4.1)
## digest        0.6.12    2017-01-27 CRAN (R 3.4.1)
## ggplot2       * 2.2.1     2016-12-30 CRAN (R 3.4.1)
## graphics      * 3.4.1     2017-06-30 local
## grDevices     * 3.4.1     2017-06-30 local
## grid          3.4.1     2017-06-30 local
## gridExtra     2.3       2017-09-09 CRAN (R 3.4.1)
## gtable        0.2.0     2016-02-26 CRAN (R 3.4.1)
## inline        0.3.14    2015-04-13 CRAN (R 3.4.1)
## labeling      0.3       2014-08-23 CRAN (R 3.4.1)
## lattice       0.20-35   2017-03-25 CRAN (R 3.4.1)
## lazyeval      0.2.0     2016-06-12 CRAN (R 3.4.1)
## magrittr      1.5       2014-11-22 CRAN (R 3.4.1)
## MASS          7.3-47    2017-02-26 CRAN (R 3.4.1)
## Matrix        1.2-10    2017-05-03 CRAN (R 3.4.1)
## methods       * 3.4.1     2017-06-30 local
## munsell       0.4.3     2016-02-13 CRAN (R 3.4.1)
## plyr          1.8.4     2016-06-08 CRAN (R 3.4.1)
## R6            2.2.2     2017-06-17 CRAN (R 3.4.1)
## RColorBrewer  1.1-2     2014-12-07 CRAN (R 3.4.1)
## Rcpp          0.12.12   2017-07-15 CRAN (R 3.4.1)
## RcppEigen     0.3.3.3.0 2017-05-01 CRAN (R 3.4.1)
## reshape2     1.4.2     2016-10-22 CRAN (R 3.4.1)
## rlang         0.1.2     2017-08-09 CRAN (R 3.4.1)
## rstan         * 2.16.2    2017-07-03 CRAN (R 3.4.1)
## scales       0.5.0     2017-08-24 CRAN (R 3.4.1)
## StanHeaders  * 2.16.0-1  2017-07-03 CRAN (R 3.4.1)
## stats         * 3.4.1     2017-06-30 local
## stats4        3.4.1     2017-06-30 local
## stringi       1.1.5     2017-04-07 CRAN (R 3.4.1)
## stringr       1.2.0     2017-02-18 CRAN (R 3.4.1)
## tibble        1.3.4     2017-08-22 CRAN (R 3.4.1)
## tools         3.4.1     2017-06-30 local
## utils         * 3.4.1     2017-06-30 local
## viridisLite   0.2.0     2017-03-24 CRAN (R 3.4.1)
```

---

## 5 References

- Baum, Leonard E, and Ted Petrie. 1966. “Statistical Inference for Probabilistic Functions of Finite State Markov Chains.” *The Annals of Mathematical Statistics* 37 (6). JSTOR: 1554–63.
- Baum, Leonard E, and George Sell. 1968. “Growth Transformations for Functions on Manifolds.” *Pacific Journal of Mathematics* 27 (2). Mathematical Sciences Publishers: 211–27.
- Baum, Leonard E. 1972. “An Inequality and Associated Maximaization Technique in Statistical Estimation for Probabilistic Functions of Markov Process.” *Inequalities* 3: 1–8.
- Baum, Leonard E., and J. A. Eagon. 1967. “An Inequality with Applications to Statistical Estimation for Probabilistic Functions of Markov Processes and to a Model for Ecology.” *Bulletin of the American Mathematical Society* 73 (3). American Mathematical Society (AMS): 360–64. doi:[10.1090/s0002-9904-1967-11751-8](https://doi.org/10.1090/s0002-9904-1967-11751-8).
- Baum, Leonard E., Ted Petrie, George Soules, and Norman Weiss. 1970. “A Maximization Technique Occurring in the Statistical Analysis of Probabilistic Functions of Markov Chains.” *The Annals of Mathematical Statistics* 41 (1). Institute of Mathematical Statistics: 164–71. doi:[10.1214/aoms/1177697196](https://doi.org/10.1214/aoms/1177697196).
- Bengio, Yoshua, and Paolo Frasconi. 1995. “An Input Output Hmm Architecture.”
- Betancourt, Michael. 2017. “Identifying Bayesian Mixture Models.” *Stan Case Studies*, no. Volume 4. [http://mc-stan.org/users/documentation/case-studies/identifying\\_mixture\\_models.html](http://mc-stan.org/users/documentation/case-studies/identifying_mixture_models.html).
- Carpenter, Bob, Andrew Gelman, Matt Hoffman, Daniel Lee, Ben Goodrich, Michael Betancourt, Michael A Brubaker, Jiqiang Guo, Peter Li, and Allen Riddell. 2016. “Stan: A Probabilistic Programming Language.” *Journal of Statistical Software* 20.
- Casella, George, and Roger L Berger. 2002. *Statistical Inference*. Vol. 2. Duxbury Pacific Grove, CA.
- Gelman, Andrew, and Donald B Rubin. 1992. “Inference from Iterative Simulation Using Multiple Sequences.” *Statistical Science*. JSTOR, 457–72.
- Jordan, Michael I. 2003. “An Introduction to Probabilistic Graphical Models.” preparation.
- Murphy, Kevin P. 2012. *Machine Learning*. MIT Press Ltd.
- Rabiner, Lawrence R. 1990. “A Tutorial on Hidden Markov Models and Selected Applications in Speech Recognition.” In *Readings in Speech Recognition*, 267–96. Elsevier. doi:[10.1016/b978-0-08-051584-7.50027-9](https://doi.org/10.1016/b978-0-08-051584-7.50027-9).
- Stan Development Team. 2017a. “RStan: The R Interface to Stan.” <http://mc-stan.org/>.
- . 2017b. “Shinystan: Interactive Visual and Numerical Diagnostics and Posterior Analysis for Bayesian Models.” <http://mc-stan.org/>.
- . 2017c. *Stan Modeling Language: User’s Guide and Reference Manual: Version 2.15.0*.
- Viterbi, A. 1967. “Error Bounds for Convolutional Codes and an Asymptotically Optimum Decoding Algorithm.” *IEEE Transactions on Information Theory* 13 (2). Institute of Electrical; Electronics Engineers (IEEE): 260–69. doi:[10.1109/tit.1967.1054010](https://doi.org/10.1109/tit.1967.1054010).
-